
Abstract Description : Like other computer languages SAS R© software provides a method
to automatically execute statements during startup of a program or
session. This paper examines the names of locations chosen in the
filename and libname statements and the placement of those names
in options that enable all programs in a project to have standardized
access to format and macro catalogs, data sets of function definitions
and folders containing reusable programs and macros. It also shows
the use of the global symbol table to provide variables for document
design.

Purpose : The purpose of this paper is to examine the default values of options,
suggest naming conventions where missing, and provide both an ex-
ample autoexec and a program to test it.

Introduction

SAS software provides a facility to automatically execute statements duringOverview
startup of a program. The option name is autoexec. Its default value is
autoexec.sas, in the Windows operating system. If a file with this name is
present in the startup folder then it is %included before the program.

The autoexec file provides all programs in the project folder the same list
of asset names and locations. Names of locations are allocated with the
filename and libname statements. These location names, referred to as
filerefs and librefs respectively, are arbitrary. Whichever names are chosen,
the next step is to insert them into the corresponding option which enables
SAS software to search for a particular object within the asset.

the matrix
Options and Location Names

asset option value default object in see also
functions cmplib libref.data n/a definitions data set
formats fmtsearch librefs work formats catalog fmterr

library
macros
autocall sasautos filerefs sasautos files folder mautosource

compiled sasmstore libref n/a definitions catalog mstored

The introduction has these topics.

• style guide

• naming conventions

• definitions

• folder structure

• syntax for lists

This paper uses italics and text fonts to identify specific uses of words.style guide

italics : SAS documentation words, e.g. fileref, libref

text : SAS keyword, e.g. filename, libname

This is the list of naming conventions used in this papersnaming conventions

site : folders available to all projects are referred to with a prefix of ’site ’

project : the ’logical name’ for the folder containing *.sas programs is ’project’

I/O : other folders have a prefix according to function:
input: ’in ’, output: ’out ’,

This is a list of the terms used here.definitions

addressing : a method of referring to another folder
absolute : provide a full (direct) path to the folder using either a drive letter

or universal-naming convention (UNC),
e.g., ’c:\temp’ also known as ’hard-coded’

relative : provide an indirect reference to a sibling or parent folder
e.g., ’..\csv’ also known as ’soft-coded’

see paragraph ’folder structure’ below

directory : see folder; in SAS documentation, see ’directory-specification’

fileref : is a ’logical name’ for one or more folders; first argument of the filename statement;
the value of a fileref is also called a ’directory-specification’

folder : a folder contains files and may contain sub-folders (or child-folders);
as used in this paper folders contain text files; also called a directory

library : is one or more folders; in SAS documentation ’library’ generally refers to folder(s) con-
taining files maintained by SAS software, i.e., catalogs and data sets

libref : is a ’logical name’ for a library, first argument of the libname statement,
the value of a libref is also called a ’directory-specification’

’logical name’ : is a phrase used in SAS documentation;
here it means a variable name in the global symbol table

The assignments discussed in this paper are based on these assumptions;folder structure
your corporate culture may differ.
The purpose of this structure is to support copying of programs from a
working project to a new project and minimize changes necessary to get
the programs working.

...\project-A\
csv comma separated values

dat data files

out-pdf vendor-specific files

out-rtf vendor-specific files

sas *.sas programs

sas7b data sets, catalogs

...\project-B\ like project-A

...\SAS-site\
includes *.sas programs

macros *.sas programs

sas7b data sets, catalogs

sas-include-tests *.sas programs

folder naming
conventions

A list contains one or more items. In the catname, filename, libname andsyntax for lists
options statements a value may be a single item or a list of items which
must be enclosed in parentheses, e.g.,
item: filename project ’.’; one folder
list: options sasautos

= (project site mac sasautos); three filerefs

Programs for discovery during research

These programs were used in research for this paper.Overview

• options define value

• options groups

• sql dictionary options

• finding word in options

• getoption: fetching value

options define value The define and value options are used to display information about an
option in the options procedure.

PROC options define value option = autoexec;

define groups This program shows the groups of the options discussed here.

PROC options group = envfiles;*autoexec fmtsearch sasautos;

PROC options group = macro ;*mautosource mstored sasmstore;

PROC options group = sasfiles;*cmplib;

sql dictionary options The sql procedure offers another avenue of discovery.

PROC sql; describe table dictionary.options;

quit;

PROC sql; select distinct group

from dictionary.options;

quit;

finding word in
options

The sql procedure can be used to search for a word in either the option
name or the option description.

*name: find-word-in-options.sas;

%let word = macro;

PROC sql; select group, optname, optdesc

from dictionary.options

where index(upcase(optname),"%upcase(&word)")

or index(upcase(optdesc),"%upcase(&word)");

quit;

The getoption function can be called in open code by the %sysfunc func-getoption: fetching
value tion. There are three types of options: boolean, character and numeric.

Boolean options use the prefix ’no’ before the option name for false.
Character and numeric option values can be fetched with the keyword op-
tion. This returns the value in the form: option-name=value. This value
can be used in a macro variable assignment statement.

sysfunc getoption *name: sysfunc-getoption.sas;

**** option type character;

%let %sysfunc(getoption(autoexec,keyword));

%put echo &=autoexec;

**** option type boolean in (T=<opt-name>,F=no<opt-name>);

%put mautosource=%sysfunc(getoption(mautosource));

options nomautosource;

%put mautosource=%sysfunc(getoption(mautosource));

log: echo AUTOEXEC=C:\...\autoexec-companion\sas\autoexec.sas

5 %put mautosource=%sysfunc(getoption(mautosource));

mautosource=MAUTOSOURCE

6 options nomautosource;

7 %put mautosource=%sysfunc(getoption(mautosource));

mautosource=NOMAUTOSOURCE

Allocating variables in the global symbol table during startup

This section describes the allocations of variable names and values in theOverview
global symbol table that are used in programs.

As used in this paper, the term ’global symbol table’ has these entities.

• environment variables

• catrefs concatenate catalog names

• filerefs location names

• librefs location names

• macro definitions %let mvar=...;

• macro variables %macro ... %mend;

• options

• running text: titles and footnotes

Global variables for output

Managing output has two aspects, marking up the document and where toOverview
write the document, i.e., the destination folder.

• document design

• output destination

Output documents may have running text, which consists of either or bothdocument design
of headers or footers. The global symbol table contains a maximum of ten
entries for title or footnote assignments.
The naming conventions used here are title1 contains the project namenaming conventions→
and title2 contain the name of the program; titles 3–10 are assigned and
cleared by programs.

title1 'An autoexec.sas Companion';

*** sysin contains path\program-name.sas;

title2 "%sysfunc(getoption(sysin))";

* show program-name only;

title2 "%scan(%sysfunc(getoption(sysin)),-1,/\)";

output destination All output may be directed to a single folder with the use of an environment
variable. A bang or exclamation point (!) is used to refer to an environment
variable.

******** allocate environment variables;

*** relative: sibling folder;

options set = out_pdf '..\out-pdf';

ods pdf file = ’!out pdf\report.pdf’;usage:

Refer to the paragraph ’folder structure’ above, on page 3.naming conventions→
Other authors use macro variables for this task.commentary

Filerefs for input, programs and autocall macros

One important use of the global symbol table is for names of locations ofOverview
reusable programs and input text. This section provides suggestions for
naming of the following items.

• filerefs for input

• filerefs for programs

• autocall macros

Use ’in ’ as the prefix of names for folders containing text files of input.filerefs for input
These example shows the folders are sibling folders and are described
with relative addressing.

filename in csv ’..\csv’;

filename in dat ’..\dat’;

Refer to the paragraph ’folder structure’ above, on page 3.naming conventions→

Project programs may reuse other programs in the project folder as well asfilerefs for programs
utility programs in the site folder.
Use ’site ’ as the prefix of location names in the site folder.naming conventions→
filename project '.'; * here: startup folder;

*** allocate environment variables (e-var);

* absolute: hard-coded;

options set = site_root 'C:\SAS-projects\SAS-site';

*** use bang (!) as prefix to reference an e-var;

filename site_inc "!site_root\includes\"; * double quotes;

filename site_mac '!site_root\macros\' ; *or single quotes;

• This example shows the environment variable ’site root’ is hard-codednotes:
with a path to a folder that is not a sibling of the project folder.

• Either single or double quotes may be used.

There are a pair of options for autocall macros;autocall macros
mautosource is boolean, sasautos is character and is a list of filerefs.
This options statment sets up the autocall macro search facility for three
filerefs: project, site mac, and the SAS-supplied fileref sasautos.

** pair macro autocall: boolean, search-path filerefs;

options mautosource

sasautos = (project site_mac sasautos);

The list of filerefs must be enclosed in parentheses.note:

Catrefs and librefs for format and macro catalogs, data sets and functions

This section describes naming convention choices and statements to allo-Overview
cate location names for SAS-managed files such as catalogs of formats,
compiled macros, and data sets containing functions.

• choose libref library

• catalogs for formats

• catalogs for macros

• catrefs to concatenate catalogs

• data sets for fcmp functions

There are two reasons to use the libref library.choose libref library

• SAS uses the term throughout its documentation

• the default search path for format catalogs is library

The format procedure compiles value statements and other definitionscatalogs for formats
and saves into the the libref named in the library option of the procedure
statement.

syntax : libname libref ’folder’;

usage : libname library ’..\sas7b’; * sibling folder;

The statement proc format library = library;note:
creates a catalog in the libref library.
The catalog name is library.formats.naming conventions→
Libref library is the second libref in the default format-search path.

Examine the format search path with this statement.

%put %sysfunc(getoption(fmtsearch,keyword));

log: fmtsearch=(WORK LIBRARY)

Do not change the format search path to (library work) because:beware naming collisions

• it is frustrating for users to create formats, which are stored in the
libref work, that are superseded by formats in the libref library

• it is extremely difficult to debug this problem

The format catalogs in librefs work and library are always searched evennote:
if they are not in the fmtsearch list.

Macros may be compiled and stored in a catalog.catalogs for compiled
macros Use the macro statement option store to save in a catalog.

%macro demo(...)/store;example:

There are a pair of options for compiled and stored macros;
mstored is boolean, sasmstore is character and is one libref.

libname site_lib '!site_root\sas7b';

** pair macros compiled and stored: boolean, libref;

options mstored sasmstore = site_lib;* or library;

The name of the catalog where compiled and stored macro definitions arenote:
saved is libref.sasmacr. The search sequence is libref work where autocall
macros are saved, then libref of option sasmstore; this search sequence
cannot be changed. Compare to the option fmtsearch on page 8.
Be aware of naming collisions — masking — when using compiled andnaming conventions→
stored macros, i.e., macros with the same name in both the work and
sasmstore catalogs.

SASautos Companion [6] has a utility program, listmcat, to find naming
collisions in macro catalogs.

catrefs of multiple
catalogs

The catname statement can be used to concatenate catalogs of either for-
mats or macros.

*setup: allocate librefs;

libname lib_xtra '..\sas7b';

libname lib_site '..\..\SAS-site\sas7b';

*usage: allocate catref for two or more catalogs;

catname cat_fmts (lib_xtra.formats lib_site.formats);

options fmtsearch = (work library cat_fmts);

There are a pair of statements for functions created by the fcmp procedure.data set for fcmp
functions libname allocates a libref ; option cmplib is character;

it is a two-level data set name, e.g., libref.data.
Both the libref and data set name are arbitrary.naming conventions→
*creation: proc fcmp outlib=libref.data_set_name.package;

******* functions compiled by proc fcmp;

options cmplib = site_lib.functions; *or library.functions;

Other options

There are several hundred options many of which may be set in the au-Overview
toexec. See the paragraph Suggested Reading below for other authors’
recommendations.

options fullstimer %* max info in timer block;

msglevel = i %* more messages in log ;

nofmterr; * no err: formats missing;

Many of the above options can be modified with either or both of the optionsappend, insert
append or insert.

*syntax: options insert|append=(opt-name=libref|'..\path');

*example;

options insert=(fmtsearch=lib_a);*libref;

*discovery;

proc options option=fmtsearch value;

In SAS documentation see the pages APPEND= System Option,note:
INSERT= System Option, and Using SAS System Options,
paragraphs: Option Value Information for an Option Modified by the INSERT and APPEND
Options and Changing an Option Value by Using the INSERT and APPEND System Options.

Programs

This section contains the listings of an autoexec.sas and test programs.Overview

autoexec.sas
title1 'An autoexec.sas Companion';

%let %sysfunc(getoption(sysin,keyword));

%put echo &=sysin;

title2 "&sysin"; *** path\program-name.sas;

title2 "%scan(&sysin,-2,/\.)"; *** program-name only;

*** allocate environment variables;

*** relative: sibling folder;

options set = out_pdf "..\out-pdf"; * double quotes;

options set = out_rtf '..\out-rtf'; *or single quotes;

*** absolute: hard-coded compare to site_lib;

options set = site_root 'C:\SAS-projects\SAS-site';

*** various folders with input files;

filename in_csv '..\csv';

filename in_dat '..\dat';

filename project '.';* here: startup folder: *.sas programs;

*** use bang (!) as prefix to reference an e-var;

filename site_inc "!site_root\includes"; * double quotes;

filename site_mac '!site_root\macros' ; *or single quotes;

libname library '..\sas7b'; *sibling folder;

*libname site_lib '!site_root\sas7b';

libname site_lib '..\..\SAS-site\sas7b';*compare to site_root;

*catname catref ...;

*** functions compiled by proc fcmp in data set;

options cmplib = library.functions;*or site_lib.functions;

*** format catalogs, see also option fmterr;

options fmtsearch = (work library site_lib.formats);

** pair macro autocall: boolean, search-path filerefs;

options mautosource

sasautos = (project site_mac sasautos);

** pair macros compiled and stored: boolean, one libref;

options mstored sasmstore = library; *or site_lib;

options nofmterr %* no err msg: fmt missing;

Autoexec-test.*

The autoexec-test program uses the following functions.Overview

cexist : exist catalog

exist : exist libref.data-set

getoption : returns value of an option

ifc : evaluate condition, return character string

%nrstr : no rescan string, delays evaluation of references in argument

%sysfunc : system function, allows access to certain data step functions;
used here with ifc to conditionally execute statements

%sysfunc(ifc : %sysfunc(ifc(condition

,%nrstr(%put true ;)

,%nrstr(%put false;) ))

evaluate a condition in open code, return either true statement(s) or
false statement(s);
the %nrstr function guarantees only one statement is returned

%let catalog= : macro variable assignment statement

exist catalog : %let exist catalog = %nrstr(echo

%sysfunc(ifc(%sysfunc(cexist(&catalog))

,,%nrstr(not ),))exist &catalog);

this macro variable contains a function call to verify the existence of
the catalog named in the macro variable catalog

%unquote : %let catalog = work.formats;

%put %unquote(&exist catalog);

resolves the macro variable references and function calls in the macro
variable exist catalog

sas.cmd This batch file calls SAS software.

rem name: sas.cmd

call "C:\Program Files\SASHome\SASFoundation\9.4\sas.exe" %*

The DOS command ’call’ allows multiple instances in another batch file.
The string percent+asterisk ’%*’ passes all command-line parameters on
to sas.exe.
See usage in autoexec-test.bat below.

sasv9.cfg
/* name: <project-name>\sas\sasv9.cfg */

-sasinitialfolder '.' /* set 'file-open' to startup folder */

Without this project configuration file the File-Open dialogue defaults to
"C:\Documents and Settings\<user-name>\My Documents\My SAS Files\9.4"

autoexec-test.bat This is the batch file used to test an autoexec. The startup-only option
echoauto is used to echo the statements in the autoexec;
its action is similar to %include ’autoexec.sas’\source2;
rem name: autoexec-test.bat

sas autoexec-test -echoauto

autoexec-test.sas **** name: autoexec-test.sas;

**** many options are set in configuration files;

%put echo sascfg=%sysget(sascfg);

%put echo %sysfunc(getoption(config,keyword));

*command-line option sasinitialfolder may be set by icons;

*or project config, compare to folder of fileref project;

%put echo %sysfunc(getoption(sasinitialfolder,keyword));

*review all filerefs;

filename _all_ list;

**** review cmplib=libref.data-set-name of functions;

%let %sysfunc(getoption(cmplib,keyword));

%sysfunc(ifc(%sysfunc(exist(&cmplib))

,%nrstr(%put echo exist(&cmplib);)

,%nrstr(%put echo not exist(&cmplib);),))

%let exist_catalog = %nrstr(echo

%sysfunc(ifc(%sysfunc(cexist(&catalog))

,,%nrstr(not ),))exist &catalog);

**** formats: search list of librefs;

%let %sysfunc(getoption(fmtsearch,keyword));*remove parens;

%let fmtsearch=%substr(&fmtsearch,2,%length(&fmtsearch)-2);

%put echo &=fmtsearch;

**** check default = work library;

%let catalog = work.formats;

%put %unquote(&exist_catalog);

%let catalog = library.formats;

%put %unquote(&exist_catalog);

**** remove default librefs WORK and LIBRARY;

%let fmtsearch=%sysfunc(tranwrd(&fmtsearch,WORK,%str( )));

%let fmtsearch=%sysfunc(tranwrd(&fmtsearch,LIBRARY,%str( )));

**** there may be one, or more;

%let fmt_lib = %scan(&fmtsearch,1,%str( ));

**** if fmt-lib has dot then is two-level name;

%let catalog = %sysfunc(ifc(%index(&fmt_lib,.)

,%nrstr(&fmt_lib),%nrstr(&fmt_lib..formats) ));

%put %unquote(&exist_catalog);

**** macro autocall;

%put echo mautosource=%sysfunc(getoption(mautosource));

%put echo %sysfunc(getoption(sasautos,keyword));

*allocate fileref sasautos to check if modified in config;

%include sasautos(af)/nosource2;

%put echo sasroot=%sysget(sasroot);

filename sasautos list;

**** if sasautos contains paths other than sasroot

then review configuration file(s) listed above;

**** macros compiled and stored;

%let mstored =%sysfunc(getoption(mstored));

%let sasmstore=%sysfunc(ifc(&mstored eq MSTORED

,%nrstr(%sysfunc(getoption(sasmstore))),%nrstr(work)));

%put echo &=mstored &=sasmstore;

%let catalog = &sasmstore..sasmacr;

%put %unquote(&exist_catalog);

autoexec-test-log.txt NOTE: AUTOEXEC processing beginning; file is
C:/.../autoexec-companion/sas/autoexec.sas.

19 libname library '..\sas7b';
NOTE: Libref LIBRARY was successfully assigned as follows:

Physical Name: C:\...\autoexec-companion\sas7b
NOTE: AUTOEXEC processing completed.
1 **** name: autoexec-test.sas;
2 **** many options are set in configuration files;
3 %put echo sascfg=%sysget(sascfg);
echo: sascfg=C:\...\SASHome\SASFoundation\9.4\nls\en
4 %put echo %sysfunc(getoption(config,keyword));
echo: CONFIG=("C:\...\SASHome\SASFoundation\9.4\sasv9.cfg"

"C:\...\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"
"C:\...\autoexec-companion\sas\sasv9.cfg" )

8 %put echo %sysfunc(getoption(sasinitialfolder,keyword));
echo: SASINITIALFOLDER=.
11 filename _all_ list;
...
NOTE: Fileref= PROJECT

Physical Name= C:\...\autoexec-companion\sas
NOTE: Fileref= SITE_INC

Physical Name= C:\SAS-projects\SAS-site\includes
NOTE: Fileref= SITE_MAC

Physical Name= C:\SAS-projects\SAS-site\macros
13 **** review cmplib=libref.data-set-name of functions;
14 %let %sysfunc(getoption(cmplib,keyword));
15 %sysfunc(ifc(%sysfunc(exist(&cmplib))
16 ,%nrstr(%put echo exist(&cmplib);)
17 ,%nrstr(%put echo not exist(&cmplib);),))
echo: not exist(library.functions)
19 %let exist_catalog = %nrstr(echo
20 %sysfunc(ifc(%sysfunc(cexist(&catalog))
21 ,,%nrstr(not ),))exist &catalog);
23 **** formats: search list of librefs;
24 %let %sysfunc(getoption(fmtsearch,keyword));*remove parens;
25 %let fmtsearch=%substr(&fmtsearch,2,%length(&fmtsearch)-2);
26 %put echo &=fmtsearch;
echo: FMTSEARCH=WORK LIBRARY SITE_LIB.FORMATS
27 **** check default = work library;
28 %let catalog = work.formats;
29 %put %unquote(&exist_catalog);
echo: not exist work.formats
30 %let catalog = library.formats;
31 %put %unquote(&exist_catalog);
echo: not exist library.formats
32 **** remove default librefs WORK and LIBRARY;
33 %let fmtsearch=%sysfunc(tranwrd(&fmtsearch,WORK,%str( )));
34 %let fmtsearch=%sysfunc(tranwrd(&fmtsearch,LIBRARY,%str( )));
35 **** there may be one, or more;
36 %let fmt_lib = %scan(&fmtsearch,1,%str( ));
37 **** if fmt-lib has dot then is two-level name;
38 %let catalog = %sysfunc(ifc(%index(&fmt_lib,.)
39 ,%nrstr(&fmt_lib),%nrstr(&fmt_lib..formats) ));
40 %put %unquote(&exist_catalog);
echo: not exist SITE_LIB.FORMATS
42 **** macro autocall;
43 %put echo mautosource=%sysfunc(getoption(mautosource));
echo: mautosource=MAUTOSOURCE
44 %put echo %sysfunc(getoption(sasautos,keyword));
echo: SASAUTOS=(project site_mac sasautos)
...
72 %put echo sasroot=%sysget(sasroot);
echo: sasroot=C:\...\SASHome\SASFoundation\9.4
73 filename sasautos list;
NOTE: Fileref= SASAUTOS

Physical Name= C:\...\SASHome\SASFoundation\9.4\core\sasmacro
...

Physical Name= C:\...\SASHome\SASFoundation\9.4\stat\sasmacro
77 **** macros compiled and stored;
78 %let mstored =%sysfunc(getoption(mstored));
79 %let sasmstore=%sysfunc(ifc(&mstored eq MSTORED
80 ,%nrstr(%sysfunc(getoption(sasmstore))),%nrstr(work) ));
81 %put echo &=mstored &=sasmstore;
echo: MSTORED=MSTORED SASMSTORE=library
82 %let catalog = &sasmstore..sasmacr;
83 %put %unquote(&exist_catalog);
echo: not exist library.sasmacr

Summary

An autoexec file provides the ability to centralize statements which manageConclusion
entries into and relations within the global symbol table. This centralization
encourages standardization and reduces the complexity, and therefore the
testing burden, of all programs in the project.

Suggested Reading

